Lie Groups and Differential Equations

or How I Learned to Stop Worrying and Love the Symmetries

İsmail Deniz Gün

Boğaziçi University

July 21, 2024

Outline

Motivation

- Ever wondered how they find the symmetries of the equations you study in your classes other than just staring at it really hard?
- Ever encountered a rather quarrelsome equation and wished you had the tools to tame it?

One Parameter Group of Point Transformations

Definition: $\Gamma_\epsilon : \mathbb{R}^n \to \mathbb{R}^n$ is a one parameter group of point transformations if:

1. $\Gamma_0(x_1, \ldots, x_n) = (x_1, \ldots, x_n)$
2. $\Gamma_\epsilon \circ \Gamma_\delta = \Gamma_{\delta+\epsilon}$
3. Γ has a MacLaurin series with respect to ϵ (i.e., Taylor expansion around zero, i.e., smooth at zero, i.e., infinitely differentiable, etc.)

Examples

- Transformation group example:

$$\Gamma_\epsilon(x, y) = (x\cos(\epsilon) + y\sin(\epsilon), -x\sin(\epsilon) + y\cos(\epsilon))$$

- Represented with a system of equations:

$$\begin{cases} \bar{x} = x\cos(\epsilon) + y\sin(\epsilon) \\ \bar{y} = -x\sin(\epsilon) + y\cos(\epsilon) \end{cases}$$

- Another example of a transformation:

$$\Gamma_\epsilon(x, y) = (x + \epsilon, y)$$

Infinitesimal Generators

- A basis for the tangent space of the identity element in the group.
- Every element in the connected component of the group can be "generated" by repeatedly applying these transformations [1].

[1] Spencer, "What is the Lie group infinitesimal generator?", Math Stack Exchange. Accessed: July 2024.

Infinitesimal Generators:

$$\bar{x}_1 = \bar{x}_1(x_1, x_2, \ldots, x_n; \epsilon)$$
$$\bar{x}_2 = \bar{x}_2(x_1, x_2, \ldots, x_n; \epsilon)$$
$$\vdots$$
$$\bar{x}_n = \bar{x}_n(x_1, x_2, \ldots, x_n; \epsilon)$$

$$\xi_1 = \left.\frac{\partial \bar{x}_1}{\partial \epsilon}\right|_{\epsilon=0} \quad \xi_2 = \left.\frac{\partial \bar{x}_2}{\partial \epsilon}\right|_{\epsilon=0} \quad \cdots \quad \xi_n = \left.\frac{\partial \bar{x}_n}{\partial \epsilon}\right|_{\epsilon=0}$$

$$\bar{x}_1 = x_1 + \epsilon\xi_1 + O(\epsilon^2)$$
$$\bar{x}_2 = x_2 + \epsilon\xi_2 + O(\epsilon^2)$$
$$\vdots$$
$$\bar{x}_n = x_n + \epsilon\xi_n + O(\epsilon^2)$$

Infinitesimal Generators:

- So to the first order:

$$X = \xi_1 \frac{\partial}{\partial x_1} + \xi_2 \frac{\partial}{\partial x_2} + \ldots + \xi_n \frac{\partial}{\partial x_n}$$

Examples:

$$\Gamma_\epsilon(x,y) = (x\cos(\epsilon) + y\sin(\epsilon), -x\sin(\epsilon) + y\cos(\epsilon))$$

$$\begin{cases} \bar{x_1} = x_1\cos(\epsilon) + x_2\sin(\epsilon) \\ \bar{x_2} = -x_1\sin(\epsilon) + x_2\cos(\epsilon) \end{cases}$$

$$\xi_1 = \frac{\partial \bar{x}_1}{\partial \epsilon}\bigg|_{\epsilon=0} = -x_1\sin(\epsilon) + x_2\cos(\epsilon)\bigg|_{\epsilon=0} = x_2$$

$$\xi_2 = \frac{\partial \bar{x}_2}{\partial \epsilon}\bigg|_{\epsilon=0} = -x_1\cos(\epsilon) - x_2\sin(\epsilon)\bigg|_{\epsilon=0} = -x_1$$

Examples:

$$X = x_2 \frac{\partial}{\partial x_1} - x_1 \frac{\partial}{\partial x_2}$$

- Which you might have seen as:

$$X_g = \begin{pmatrix} 0 & -1 \\ 1 & 0 \end{pmatrix}$$

Examples:

$$\Gamma_\epsilon(x, y) = (x + \epsilon, y)$$

$$\bar{x} = x + \epsilon$$

$$\bar{y} = y$$

$$\xi_1 = \left.\frac{\partial \bar{x}}{\partial \epsilon}\right|_{\epsilon=0} = 1$$

$$\xi_2 = \left.\frac{\partial \bar{y}}{\partial \epsilon}\right|_{\epsilon=0} = 0$$

Examples:

$$X = \frac{\partial}{\partial x}$$

- Which is just what generates the x-axis translations.
- Can we also find Γ_ϵ from X?

$$X = \xi_1 \frac{\partial}{\partial x_1} + \xi_2 \frac{\partial}{\partial x_2} + \ldots + \xi_n \frac{\partial}{\partial x_n}$$

$$\begin{cases} \dfrac{\partial \bar{x_1}}{\partial \epsilon} = \xi_1(\bar{x_1}, \bar{x_2}, \ldots, \bar{x_n}) \\ \qquad\qquad\qquad\qquad\qquad\qquad \vdots \\ \dfrac{\partial \bar{x_n}}{\partial \epsilon} = \xi_n(\bar{x_1}, \bar{x_2}, \ldots, \bar{x_n}) \end{cases}$$

Example:

$$X = 2x\frac{\partial}{\partial x} + (y+1)\frac{\partial}{\partial y}$$

$$\xi_1 = 2x \quad \xi_2 = y + 1$$

$$\frac{\partial \bar{x}}{\partial \epsilon} = 2\bar{x}, \quad \bar{x}\bigg|_{\epsilon=0} = x, \quad \bar{x}(x, y; \epsilon)$$

$$\frac{\partial \bar{y}}{\partial \epsilon} = \bar{y} + 1, \quad \bar{y}\bigg|_{\epsilon=0} = y, \quad \bar{y}(x, y; \epsilon)$$

- We have to solve the partial differential equations for $\bar{x}$ and $\bar{y}$.

$$\frac{\partial \bar{x}}{\partial \epsilon} = 2\bar{x}, \quad \ln \bar{x} = 2\epsilon + C, \quad \bar{x} = C_1(x, y)e^{2\epsilon}$$

- At $\epsilon = 0$, $C_1(x, y) = x$. So $C_1(x, y) = x$.

$$\bar{x} = xe^{2\epsilon}$$

$$\frac{\partial \bar{y}}{\partial \epsilon} = \bar{y} + 1, \quad \ln \bar{y} + 1 = \epsilon + C, \quad \bar{y} + 1 = C_2(x, y)e^{\epsilon}$$

$$y = C_2(x, y)e^{\epsilon} - 1$$

- At $\epsilon = 0$, $C_2(x, y) = y + 1$. So $C_2(x, y) = y + 1$.

$$\bar{y} = (y + 1)e^{\epsilon} - 1$$

Example:

$$X = 2x\frac{\partial}{\partial x} + (y+1)\frac{\partial}{\partial y}$$

$$\Gamma_\epsilon(x, y) = (xe^{2\epsilon}, (y+1)e^\epsilon - 1)$$

Prolongation

We have $\Gamma_\epsilon : \mathbb{R}^2 \to \mathbb{R}^2$.
Let

$$\begin{cases} \bar{y} = \bar{y}(x, y; \epsilon) \\ \bar{x} = \bar{x}(x, y; \epsilon) \end{cases}$$

be a one parameter group of point transformations.

Prolongation

$$\frac{d\bar{y}}{d\bar{x}} = \bar{y'} = \bar{y'}(x, y, y'; \epsilon)$$

$$\frac{d^2\bar{x}}{d\bar{x}^2} = \bar{y''} = \bar{y''}(x, y, y', y''; \epsilon)$$

$$\vdots$$

is how we reach the second prolongation of Γ_ϵ. (Which will be necessary with higher order differential equations.)

A quick example

$$\underline{\Gamma_\epsilon} : \begin{cases} \bar{x} = x\cos(\epsilon) + y\sin(\epsilon) \\ \bar{y} = -x\sin(\epsilon) + y\cos(\epsilon) \end{cases}$$

$$\bar{y'} = \frac{d\bar{y}}{d\bar{x}} = \frac{d(-x\sin(\epsilon) + y\cos(\epsilon))}{d(x\cos(\epsilon) + y\sin(\epsilon))} = \frac{-\sin(\epsilon)dx + \cos(\epsilon)dy}{\cos(\epsilon)dx + \sin(\epsilon)dy}\frac{\frac{1}{dx}}{\frac{1}{dx}}$$

$$= \frac{-\sin(\epsilon) + \cos(\epsilon)y'}{\cos(\epsilon) + \sin(\epsilon)y'}$$

A quick example

- The first prolongation is written as:

$$X^{(1)} = X + \eta_1 \frac{\partial}{\partial y'} \quad \eta_1 = \left. \frac{\partial y'}{\partial \epsilon} \right|_{\epsilon=0}$$

So for us:

$$\Gamma_\epsilon^1 = \begin{cases} \bar{x} = x\cos(\epsilon) + y\sin(\epsilon) \\ \bar{y} = -x\sin(\epsilon) + y\cos(\epsilon) \\ \bar{y'} = \dfrac{-\sin(\epsilon) + \cos(\epsilon)y'}{\cos(\epsilon) + \sin(\epsilon)y'} \end{cases}$$

$$\eta_1 = -1 - y'^2$$

A quick example

$$\underline{\Gamma_\epsilon} : \begin{cases} \bar{x} = x\cos(\epsilon) + y\sin(\epsilon) \\ \bar{y} = -x\sin(\epsilon) + y\cos(\epsilon) \end{cases}$$

$$\boxed{X^{(1)} = y\frac{\partial}{\partial x} - x\frac{\partial}{\partial y} + (-1 - y'^2)\frac{\partial}{\partial y'}}$$

Prolongation

- The general formula for η_1 (The factor in front of the $\frac{\partial}{\partial y'}$ term) is:

$$\bar{x} = x + \xi(x, y)\epsilon + \dots$$

$$\bar{y} = y + \eta(x, y)\epsilon + \dots$$

$$\boxed{\eta_1 = D_x\eta - y' D_x\xi}$$

where

$$D_x\eta = \frac{d}{dx}(\eta(x, y(x)))$$

- Together their general formula can be written as:

$$\boxed{X^{(2)} = \xi \frac{\partial}{\partial x} + \eta \frac{\partial}{\partial y} + \eta_1 \frac{\partial}{\partial y'} + \eta_2 \frac{\partial}{\partial y''}}$$

$$\boxed{\eta_1 = \eta_x + y'(\eta_y - \xi_x) + y'^2 \xi_y}$$

$$\boxed{\begin{aligned}\eta_2 = \eta_{xx} + y'\eta_{xy} + y''(\eta_y - \xi_x) + y'(\eta_{xy} + y'\eta_{yy} - \xi_{xx} - y'\xi_{xy}) \\ +2y'y''\xi_y + y'^2(\xi_{yx} + y'\xi_{yy}) - y''(\xi_x + y'\xi_y)\end{aligned}}$$

where subscripts denote partial derivatives.

Symmetries

- Write your ordinary differential equation as:

$$F(x, y, y', \ldots, y^{(n)}) = 0 \tag{1}$$

- If X is a symmetry of the above equation, then:

$$\text{X is a symmetry of (1)} \Leftrightarrow X^{(n)}(F)\Big|_{F=0} = 0$$

First Order ODEs

y' = w(x,y)

F(x,y,y') = w(x,y)-y'

$$X^{(1)} = \xi(x,y)\frac{\partial}{\partial x} + \eta(x,y)\frac{\partial}{\partial y} + \eta_1(x,y)\frac{\partial}{\partial y'}$$

Symmetry Condition:

$$X^{(1)}(F)\Big|_{F=0} = 0$$

$$\xi\frac{\partial}{\partial x}(w - y') + \eta\frac{\partial}{\partial y}(w - y') + \eta_1\frac{\partial}{\partial y'}(w - y') = 0$$

First Order ODEs

$$\xi w_x + \eta w_y - \eta_1 = 0$$

$$\boxed{\xi w_x + \eta w_y = \eta_1} \quad \text{for } y' = w(x, y)$$

First Order ODEs

$$\eta_1 = \eta_x + y'(\eta_y - \xi_x) + y'^2\xi_y = \xi w_x + \eta w_y$$

This gives:

$$\boxed{\xi w_x + \eta w_y = \eta_x + w(\eta_y - \xi_x) + w^2\xi_y}$$

- This is what we would be dealing with for a first order ODE.

Second Order ODEs

y''=w(x,y,y')

F(x,y,y',y'') = w(x,y,y')-y''

$$X^{(2)} = \xi(x,y)\frac{\partial}{\partial x} + \eta(x,y)\frac{\partial}{\partial y} + \eta_1(x,y)\frac{\partial}{\partial y'} + \eta_2(x,y)\frac{\partial}{\partial y''}$$

Symmetry Condition:

$$X^{(2)}(F)\Big|_{F=0} = 0$$

$$\boxed{\xi w_x + \eta w_y + \eta_1 w_{y'} = \eta_2}$$

Second Order ODEs

$$\eta_1 = \eta_x + y'(\eta_y - \xi_x) + y'^2\xi_y$$

$$\eta_2 = \eta_{xx} + y'\eta_{xy} + y''(\eta_y - \xi_x) + y'(\eta_{xy} + y'\eta_{yy} - \xi_{xx} - y'\xi_{xy}) + 2y'y''\xi_y + y'^2(\xi_{yx} + y'\xi_{yy}) - y''(\xi_x + y'\xi_y)$$

where we can replace y" with w(x,y,y').

$$\boxed{\begin{aligned} &\xi w_x + \eta w_y + (\eta_x + y'(\eta_y - \xi_x) + y'^2\xi_y)w_{y'} \\ &= \eta_{xx} + y'\eta_{xy} + w(\eta_y - \xi_x) + y'(\eta_{xy} + y'\eta_{yy} - \xi_{xx} - y'\xi_{xy}) \\ &+2y'w\xi_y + y'^2(\xi_{yx} + y'\xi_{yy}) - w(\xi_x + y'\xi_y) \end{aligned}}$$

Second Order ODEs

- Unknowns:

$$\xi(x, y) \quad \eta(x, y)$$

- With the symmetry condition, we will equate the powers of y' to zero:

$$(A_1) + y'(A_2) + y'^2(A_3) + y'^3(A_4) + \ldots = 0$$

$$A_1 = 0$$

$$A_2 = 0$$

$$A_3 = 0$$

$$\vdots$$

Second Order ODEs

$$
\boxed{\begin{aligned}
\xi w_x + \eta w_y &= \eta_{xx} + w(\eta_y - \xi_x) + w\xi_x \\
0 &= \eta_{xy} + \eta_{xy} - \xi_{xx} + 2w\xi_y - wy' \\
0 &= \eta_{yy} \\
0 &= \xi_{yy}
\end{aligned}}
$$

An example

- $y'' = x^3 y^2$

$$\eta_{yy} = 0 \longrightarrow \eta = a_1(x) + a_2(x)y$$

$$\xi_{yy} = 0 \longrightarrow \xi = b_1(x) + b_2(x)y$$

Solving the zeroth and first power equations then gives:

$$\xi = 0 \quad \eta = C_1 x + C_2$$

giving the generator:

$$X = 0.\frac{\partial}{\partial x} + (C_1 x + C_2)\frac{\partial}{\partial y}$$

Reduction of Order

- Finding the symmetries of a certain equation can allow us to see what types of freedoms we have in our system (i.e. translational or rotational invariance etc.)
- But are there any practical applications of this? (i.e. What is the spoon that comes out of all this woodworking?)

Reduction of Order Example

$$y'' = 3(x - y)(y')^3$$

$$X = (x - y)\frac{\partial}{\partial x}$$

- Using the symmetry let's transform:

$$(x, y) \longrightarrow (t, s)$$

where we'd rather not have "t" explicit in the final equation.

Reduction of Order Example

$$Xt = 0, \quad (x-y)t_x = 0 \to t = C_1 t(y)$$
$$Xs = 1, \quad (x-y)s_x = 1$$

$$\int ds = \int \frac{dx}{x-y} \to s = \ln(x-y) + C_2 s(y)$$

With this transformation we have some freedom in coefficients.
Let's choose

$$\begin{cases} t = y \\ s = \ln(x-y) + y \end{cases}$$

$$y = t$$
$$x = e^{s-t} + t$$

Reduction of Order Example

- Now find y' and y" and insert back:

$$y' = \frac{dy}{dx} = \frac{dt}{d(e^{s-t} + t)}$$

$$\boxed{y' = \frac{1}{e^{s-t}[\dot{s} - 1] + 1}}$$

$$\boxed{y'' = \frac{e^{s-t}[(\dot{s} - 1)(1 - \dot{s}) - \ddot{s}]}{[e^{s-t}(\dot{s} - 1) + 1]^3}}$$

Reduction of Order Example

$$y'' = 3(x-y)(y')^3$$

$$-e^{s-t}[\ddot{s} + (\dot{s})^2 - 2\dot{s} + 1] = 3e^{s-t}$$

$$\ddot{s} + (\dot{s})^2 - 2\dot{s} + 1 = -3$$

$$\ddot{s} + (\dot{s})^2 - 2\dot{s} + 4 = 0$$

Put $\dot{s} = w$:

$$\boxed{\dot{w} + w^2 - 2w + 4 = 0} \quad \text{Reduced order by 1!}$$

ODE symmetries and Computers

- For demonstration we will be using Mathematica but feel free to use any other type of software.

Say our ODE is:

$$y''' = yy'' - (y')^2$$

ODE symmetries and Computers

- We will begin by defining our eta_1, eta_2, and eta_3. These are the same for all ODEs, so feel free to reuse.

Finding Symmetries of $y''' = yy'' - (y')^2$

```
eta1 = D[η[x, y[x]], x] - y'[x] × D[ξ[x, y[x]], x]
```

$y'[x]\,\eta^{(0,1)}[x, y[x]] + \eta^{(1,0)}[x, y[x]] - y'[x]\left(y'[x]\,\xi^{(0,1)}[x, y[x]] + \xi^{(1,0)}[x, y[x]]\right)$

ODE symmetries and Computers

- Hint: To type ξ (xi), you can type Esc xi Esc.

$y''' = yy'' - (y')^2$

```
eta2 = D[eta1, x] - y''[x] × D[ξ[x, y[x]], x]
```

$$y''[x]\,\eta^{(0,1)}[x, y[x]] - 2\,y''[x]\left(y'[x]\,\xi^{(0,1)}[x, y[x]] + \xi^{(1,0)}[x, y[x]]\right) + y'[x]\,\eta^{(1,1)}[x, y[x]] + y'[x]\left(y'[x]\,\eta^{(0,2)}[x, y[x]] + \eta^{(1,1)}[x, y[x]]\right) + \eta^{(2,0)}[x, y[x]] - y'[x]\left(y''[x]\,\xi^{(0,1)}[x, y[x]] + y'[x]\,\xi^{(1,1)}[x, y[x]] + y'[x]\left(y'[x]\,\xi^{(0,2)}[x, y[x]] + \xi^{(1,1)}[x, y[x]]\right) + \xi^{(2,0)}[x, y[x]]\right)$$

$y''' = yy'' - (y')^2$

```
eta3 = D[eta2, x] - y'''[x] × D[ξ[x, y[x]], x]
```

$y^{(3)}[x]\,\eta^{(0,1)}[x, y[x]] - 3\,y^{(3)}[x]\,\left(y'[x]\,\xi^{(0,1)}[x, y[x]] + \xi^{(1,0)}[x, y[x]]\right) +$
$y''[x]\,\eta^{(1,1)}[x, y[x]] + 2\,y''[x]\,\left(y'[x]\,\eta^{(0,2)}[x, y[x]] + \eta^{(1,1)}[x, y[x]]\right) -$
$3\,y''[x]\,\left(y''[x]\,\xi^{(0,1)}[x, y[x]] + y'[x]\,\xi^{(1,1)}[x, y[x]] +\right.$
$\left.y'[x]\,\left(y'[x]\,\xi^{(0,2)}[x, y[x]] + \xi^{(1,1)}[x, y[x]]\right) + \xi^{(2,0)}[x, y[x]]\right) +$
$y'[x]\,\eta^{(2,1)}[x, y[x]] + y'[x]\,\left(y'[x]\,\eta^{(1,2)}[x, y[x]] + \eta^{(2,1)}[x, y[x]]\right) +$
$y'[x]\,\left(y''[x]\,\eta^{(0,2)}[x, y[x]] + y'[x]\,\eta^{(1,2)}[x, y[x]] +\right.$
$\left.y'[x]\,\left(y'[x]\,\eta^{(0,3)}[x, y[x]] + \eta^{(1,2)}[x, y[x]]\right) + \eta^{(2,1)}[x, y[x]]\right) +$
$\eta^{(3,0)}[x, y[x]] - y'[x]\,\left(y^{(3)}[x]\,\xi^{(0,1)}[x, y[x]] + y''[x]\,\xi^{(1,1)}[x, y[x]] +\right.$
$2\,y''[x]\,\left(y'[x]\,\xi^{(0,2)}[x, y[x]] + \xi^{(1,1)}[x, y[x]]\right) +$
$y'[x]\,\xi^{(2,1)}[x, y[x]] + y'[x]\,\left(y'[x]\,\xi^{(1,2)}[x, y[x]] + \xi^{(2,1)}[x, y[x]]\right) +$
$y'[x]\,\left(y''[x]\,\xi^{(0,2)}[x, y[x]] + y'[x]\,\xi^{(1,2)}[x, y[x]] +\right.$
$\left.\left.y'[x]\,\left(y'[x]\,\xi^{(0,3)}[x, y[x]] + \xi^{(1,2)}[x, y[x]]\right) + \xi^{(2,1)}[x, y[x]]\right) + \xi^{(3,0)}[x, y[x]]\right)$

$y''' = yy'' - (y')^2$

```
rrule = {y[x] → Y0, y'[x] → Y1, y''[x] → Y2, y'''[x] → Y3};
```

Replace[eta1, rrule] and eta1 /. rrule are same

```
Eta1 = eta1 /. rrule
Eta2 = eta2 /. rrule
Eta3 = eta3 /. rrule
```

$$\mathrm{Y1}\,\eta^{(0,1)}[x,\mathrm{Y0}] + \eta^{(1,0)}[x,\mathrm{Y0}] - \mathrm{Y1}\left(\mathrm{Y1}\,\xi^{(0,1)}[x,\mathrm{Y0}] + \xi^{(1,0)}[x,\mathrm{Y0}]\right)$$

$$\begin{aligned}&\mathrm{Y2}\,\eta^{(0,1)}[x,\mathrm{Y0}] - 2\,\mathrm{Y2}\left(\mathrm{Y1}\,\xi^{(0,1)}[x,\mathrm{Y0}] + \xi^{(1,0)}[x,\mathrm{Y0}]\right) + \\ &\mathrm{Y1}\,\eta^{(1,1)}[x,\mathrm{Y0}] + \mathrm{Y1}\left(\mathrm{Y1}\,\eta^{(0,2)}[x,\mathrm{Y0}] + \eta^{(1,1)}[x,\mathrm{Y0}]\right) + \eta^{(2,0)}[x,\mathrm{Y0}] - \\ &\mathrm{Y1}\left(\mathrm{Y2}\,\xi^{(0,1)}[x,\mathrm{Y0}] + \mathrm{Y1}\,\xi^{(1,1)}[x,\mathrm{Y0}] + \mathrm{Y1}\left(\mathrm{Y1}\,\xi^{(0,2)}[x,\mathrm{Y0}] + \xi^{(1,1)}[x,\mathrm{Y0}]\right) + \xi^{(2,0)}[x,\mathrm{Y0}]\right)\end{aligned}$$

$$\begin{aligned}&\mathrm{Y3}\,\eta^{(0,1)}[x,\mathrm{Y0}] - 3\,\mathrm{Y3}\left(\mathrm{Y1}\,\xi^{(0,1)}[x,\mathrm{Y0}] + \xi^{(1,0)}[x,\mathrm{Y0}]\right) + \\ &\mathrm{Y2}\,\eta^{(1,1)}[x,\mathrm{Y0}] + 2\,\mathrm{Y2}\left(\mathrm{Y1}\,\eta^{(0,2)}[x,\mathrm{Y0}] + \eta^{(1,1)}[x,\mathrm{Y0}]\right) - \\ &3\,\mathrm{Y2}\left(\mathrm{Y2}\,\xi^{(0,1)}[x,\mathrm{Y0}] + \mathrm{Y1}\,\xi^{(1,1)}[x,\mathrm{Y0}] + \mathrm{Y1}\left(\mathrm{Y1}\,\xi^{(0,2)}[x,\mathrm{Y0}] + \xi^{(1,1)}[x,\mathrm{Y0}]\right) + \xi^{(2,0)}[x,\mathrm{Y0}]\right) + \\ &\mathrm{Y1}\,\eta^{(2,1)}[x,\mathrm{Y0}] + \mathrm{Y1}\left(\mathrm{Y1}\,\eta^{(1,2)}[x,\mathrm{Y0}] + \eta^{(2,1)}[x,\mathrm{Y0}]\right) + \\ &\mathrm{Y1}\left(\mathrm{Y2}\,\eta^{(0,2)}[x,\mathrm{Y0}] + \mathrm{Y1}\,\eta^{(1,2)}[x,\mathrm{Y0}] + \mathrm{Y1}\left(\mathrm{Y1}\,\eta^{(0,3)}[x,\mathrm{Y0}] + \eta^{(1,2)}[x,\mathrm{Y0}]\right) + \eta^{(2,1)}[x,\mathrm{Y0}]\right) + \end{aligned}$$

$y''' = yy'' - (y')^2$

- Hint: Right arrow can be written with \[RightArrow]

$y''' = yy'' - (y')^2$

Our Equation :

```
F = Y3 - Y0 Y2 + Y1^2;
```

Y3-Y0 Y2+Y1^2 means Y3->Y0 Y2-Y1^2

```
X2F = D[F, x] ξ[x, Y0] + D[F, Y0] η[x, Y0] + Eta1 D[F, Y1] + Eta2 D[F, Y2] + Eta3 D[F, Y3]
```

$$
\begin{aligned}
&-Y2\,\eta[x, Y0] + Y3\,\eta^{(0,1)}[x, Y0] - 3\,Y3\left(Y1\,\xi^{(0,1)}[x, Y0] + \xi^{(1,0)}[x, Y0]\right) + \\
&\;2\,Y1\left(Y1\,\eta^{(0,1)}[x, Y0] + \eta^{(1,0)}[x, Y0] - Y1\left(Y1\,\xi^{(0,1)}[x, Y0] + \xi^{(1,0)}[x, Y0]\right)\right) + \\
&\;Y2\,\eta^{(1,1)}[x, Y0] + 2\,Y2\left(Y1\,\eta^{(0,2)}[x, Y0] + \eta^{(1,1)}[x, Y0]\right) - \\
&\;3\,Y2\left(Y2\,\xi^{(0,1)}[x, Y0] + Y1\,\xi^{(1,1)}[x, Y0] + Y1\left(Y1\,\xi^{(0,2)}[x, Y0] + \xi^{(1,1)}[x, Y0]\right) + \xi^{(2,0)}[x, Y0]\right) - \\
&\;Y0\left(Y2\,\eta^{(0,1)}[x, Y0] - 2\,Y2\left(Y1\,\xi^{(0,1)}[x, Y0] + \xi^{(1,0)}[x, Y0]\right) + Y1\,\eta^{(1,1)}[x, Y0] + \right.\\
&\qquad Y1\left(Y1\,\eta^{(0,2)}[x, Y0] + \eta^{(1,1)}[x, Y0]\right) + \eta^{(2,0)}[x, Y0] - Y1\left(Y2\,\xi^{(0,1)}[x, Y0] + \right.\\
&\qquad\quad \left.\left. Y1\,\xi^{(1,1)}[x, Y0] + Y1\left(Y1\,\xi^{(0,2)}[x, Y0] + \xi^{(1,1)}[x, Y0]\right) + \xi^{(2,0)}[x, Y0]\right)\right) + \\
&\;Y1\,\eta^{(2,1)}[x, Y0] + Y1\left(Y1\,\eta^{(1,2)}[x, Y0] + \eta^{(2,1)}[x, Y0]\right) + \\
&\;Y1\left(Y2\,\eta^{(0,2)}[x, Y0] + Y1\,\eta^{(1,2)}[x, Y0] + Y1\left(Y1\,\eta^{(0,3)}[x, Y0] + \eta^{(1,2)}[x, Y0]\right) + \eta^{(2,1)}[x, Y0]\right) + \\
&\;\eta^{(3,0)}[x, Y0] - Y1\left(Y3\,\xi^{(0,1)}[x, Y0] + Y2\,\xi^{(1,1)}[x, Y0] + 2\,Y2\left(Y1\,\xi^{(0,2)}[x, Y0] + \xi^{(1,1)}[x, Y0]\right) + \right.\\
&\qquad Y1\,\xi^{(2,1)}[x, Y0] + Y1\left(Y1\,\xi^{(1,2)}[x, Y0] + \xi^{(2,1)}[x, Y0]\right) + Y1\left(Y2\,\xi^{(0,2)}[x, Y0] + \right.\\
&\qquad\quad \left.\left. Y1\,\xi^{(1,2)}[x, Y0] + Y1\left(Y1\,\xi^{(0,3)}[x, Y0] + \xi^{(1,2)}[x, Y0]\right) + \xi^{(2,1)}[x, Y0]\right) + \xi^{(3,0)}[x, Y0]\right)
\end{aligned}
$$

$y''' = yy'' - (y')^2$

```
SymmCond = X2F /. Y3 → Y0 Y2 - Y1^2
```

$$
\begin{aligned}
&-Y2\,\eta[x, Y0] + \left(-Y1^2 + Y0\,Y2\right)\eta^{(0,1)}[x, Y0] - 3\left(-Y1^2 + Y0\,Y2\right)\left(Y1\,\xi^{(0,1)}[x, Y0] + \xi^{(1,0)}[x, Y0]\right) + \\
&\ 2\,Y1\left(Y1\,\eta^{(0,1)}[x, Y0] + \eta^{(1,0)}[x, Y0] - Y1\left(Y1\,\xi^{(0,1)}[x, Y0] + \xi^{(1,0)}[x, Y0]\right)\right) + \\
&\ Y2\,\eta^{(1,1)}[x, Y0] + 2\,Y2\left(Y1\,\eta^{(0,2)}[x, Y0] + \eta^{(1,1)}[x, Y0]\right) - \\
&\ 3\,Y2\left(Y2\,\xi^{(0,1)}[x, Y0] + Y1\,\xi^{(1,1)}[x, Y0] + Y1\left(Y1\,\xi^{(0,2)}[x, Y0] + \xi^{(1,1)}[x, Y0]\right) + \xi^{(2,0)}[x, Y0]\right) - \\
&\ Y0\left(Y2\,\eta^{(0,1)}[x, Y0] - 2\,Y2\left(Y1\,\xi^{(0,1)}[x, Y0] + \xi^{(1,0)}[x, Y0]\right) + Y1\,\eta^{(1,1)}[x, Y0] + \right. \\
&\qquad Y1\left(Y1\,\eta^{(0,2)}[x, Y0] + \eta^{(1,1)}[x, Y0]\right) + \eta^{(2,0)}[x, Y0] - Y1\left(Y2\,\xi^{(0,1)}[x, Y0] + \right. \\
&\qquad\qquad \left.\left. Y1\,\xi^{(1,1)}[x, Y0] + Y1\left(Y1\,\xi^{(0,2)}[x, Y0] + \xi^{(1,1)}[x, Y0]\right) + \xi^{(2,0)}[x, Y0]\right)\right) + \\
&\ Y1\,\eta^{(2,1)}[x, Y0] + Y1\left(Y1\,\eta^{(1,2)}[x, Y0] + \eta^{(2,1)}[x, Y0]\right) + \\
&\ Y1\left(Y2\,\eta^{(0,2)}[x, Y0] + Y1\,\eta^{(1,2)}[x, Y0] + Y1\left(Y1\,\eta^{(0,3)}[x, Y0] + \eta^{(1,2)}[x, Y0]\right) + \eta^{(2,1)}[x, Y0]\right) + \\
&\ \eta^{(3,0)}[x, Y0] - \\
&\ Y1\left(\left(-Y1^2 + Y0\,Y2\right)\xi^{(0,1)}[x, Y0] + Y2\,\xi^{(1,1)}[x, Y0] + 2\,Y2\left(Y1\,\xi^{(0,2)}[x, Y0] + \xi^{(1,1)}[x, Y0]\right) + \right. \\
&\qquad Y1\,\xi^{(2,1)}[x, Y0] + Y1\left(Y1\,\xi^{(1,2)}[x, Y0] + \xi^{(2,1)}[x, Y0]\right) + Y1\left(Y2\,\xi^{(0,2)}[x, Y0] + \right. \\
&\qquad\qquad \left.\left. Y1\,\xi^{(1,2)}[x, Y0] + Y1\left(Y1\,\xi^{(0,3)}[x, Y0] + \xi^{(1,2)}[x, Y0]\right) + \xi^{(2,1)}[x, Y0]\right) + \xi^{(3,0)}[x, Y0]\right)
\end{aligned}
$$

$y''' = yy'' - (y')^2$

```
SymmCond = Collect[SymmCond, {Y1, Y2}]
```

$$-3\,Y2^2\,\xi^{(0,1)}[x, Y0] - Y1^4\,\xi^{(0,3)}[x, Y0] + Y1^3\left(2\,\xi^{(0,1)}[x, Y0] + Y0\,\xi^{(0,2)}[x, Y0] + \eta^{(0,3)}[x, Y0] - 3\,\xi^{(1,2)}[x, Y0]\right) - Y0\,\eta^{(2,0)}[x, Y0] + Y2\left(-\eta[x, Y0] - Y0\,\xi^{(1,0)}[x, Y0] + 3\,\eta^{(1,1)}[x, Y0] - 3\,\xi^{(2,0)}[x, Y0]\right) + Y1^2\left(\eta^{(0,1)}[x, Y0] - Y0\,\eta^{(0,2)}[x, Y0] - 6\,Y2\,\xi^{(0,2)}[x, Y0] + \xi^{(1,0)}[x, Y0] + 2\,Y0\,\xi^{(1,1)}[x, Y0] + 3\,\eta^{(1,2)}[x, Y0] - 3\,\xi^{(2,1)}[x, Y0]\right) + \eta^{(3,0)}[x, Y0] + Y1\left(2\,\eta^{(1,0)}[x, Y0] - 2\,Y0\,\eta^{(1,1)}[x, Y0] + Y2\left(-Y0\,\xi^{(0,1)}[x, Y0] + 3\,\eta^{(0,2)}[x, Y0] - 9\,\xi^{(1,1)}[x, Y0]\right) + Y0\,\xi^{(2,0)}[x, Y0] + 3\,\eta^{(2,1)}[x, Y0] - \xi^{(3,0)}[x, Y0]\right)$$

Collect function reorganizes the expression by grouping together terms that involve the specified variables, in this case, Y1 and Y2.

$y''' = yy'' - (y')^2$

```
eq1 = Coefficient[SymmCond, Y2^2]
```

$-3\,\xi^{(0,1)}[x, Y0]$

```
DSolve[eq1 == 0, ξ, {x, Y0}]
```

```
{{ξ → Function[{x, Y0}, C[1][x]]}}
```

```
ξ[x_, Y0_] = a1[x]
```

```
a1[x]
```

$y''' = yy'' - (y')^2$

```
SymmCond = Collect[SymmCond, {Y1, Y2}]
```

$$Y1^3\, \eta^{(0,3)}[x, Y0] + Y2\left(-\eta[x, Y0] - Y0\, a1'[x] - 3\, a1''[x] + 3\, \eta^{(1,1)}[x, Y0]\right) + Y1^2\left(a1'[x] + \eta^{(0,1)}[x, Y0] - Y0\, \eta^{(0,2)}[x, Y0] + 3\, \eta^{(1,2)}[x, Y0]\right) - Y0\, \eta^{(2,0)}[x, Y0] + Y1\left(Y0\, a1''[x] - a1^{(3)}[x] + 3\, Y2\, \eta^{(0,2)}[x, Y0] + 2\, \eta^{(1,0)}[x, Y0] - 2\, Y0\, \eta^{(1,1)}[x, Y0] + 3\, \eta^{(2,1)}[x, Y0]\right) + \eta^{(3,0)}[x, Y0]$$

```
eq2 = Coefficient[SymmCond, Y1^3]
```

$$\eta^{(0,3)}[x, Y0]$$

```
DSolve[eq2 == 0, η, {x, Y0}]
```

$$\{\{\eta \to \text{Function}[\{x, Y0\}, C[1][x] + Y0\, C[2][x] + Y0^2\, C[3][x]]\}\}$$

$y''' = yy'' - (y')^2$

```
η[x_, Y0_] = b1[x] + Y0 b2[x] + Y0^2 b3[x]
```

$b1[x] + Y0\, b2[x] + Y0^2\, b3[x]$

```
{ξ[x, y], η[x, y]}
```

$\{a1[x],\ b1[x] + y\, b2[x] + y^2\, b3[x]\}$

```
SymmCond = Collect[SymmCond, {Y0, Y1, Y2}]
```

$$Y1^2\,(b2[x] + a1'[x] + 6\, b3'[x]) + Y2\,(-b1[x] + 3\, b2'[x] - 3\, a1''[x]) - Y0^3\, b3''[x] + Y1\,\left(6\, Y2\, b3[x] + 2\, b1'[x] + 3\, b2''[x] - a1^{(3)}[x]\right) + b1^{(3)}[x] + Y0\,\left(Y2\,(-b2[x] - a1'[x] + 6\, b3'[x]) - b1''[x] + Y1\,(a1''[x] + 6\, b3''[x]) + b2^{(3)}[x]\right) + Y0^2\,\left(-Y2\, b3[x] - 2\, Y1\, b3'[x] - b2''[x] + b3^{(3)}[x]\right)$$

```
eq3 = Coefficient[SymmCond, Y0^3]
```

$-b3''[x]$

```
b3[x_] = c1 + c2 x;
```

$y''' = yy'' - (y')^2$

```
SymmCond = Collect[SymmCond, {Y0, Y1, Y2}]
```

$$Y1^2\ (6\ c2 + b2[x] + a1'[x]) + Y2\ (-b1[x] + 3\ b2'[x] - 3\ a1''[x]) + Y0^2\ (-2\ c2\ Y1 + (-c1 - c2\ x)\ Y2 - b2''[x]) + Y1\ \left(6\ (c1 + c2\ x)\ Y2 + 2\ b1'[x] + 3\ b2''[x] - a1^{(3)}[x]\right) + b1^{(3)}[x] + Y0\ \left(Y2\ (6\ c2 - b2[x] - a1'[x]) + Y1\ a1''[x] - b1''[x] + b2^{(3)}[x]\right)$$

```
eq3 = Coefficient[SymmCond, Y0^2]
```

$$-2\ c2\ Y1 + (-c1 - c2\ x)\ Y2 - b2''[x]$$

```
c2 = 0; c1 = 0;
b2[x_] = c3 x + c4;
```

$y''' = yy'' - (y')^2$

```
SymmCond = Collect[SymmCond, {Y0, Y1, Y2}]
```

Y1^2 (c4 + c3 x + a1′[x]) + Y2 (3 c3 - b1[x] - 3 a1″[x]) +
 Y0 (Y2 (-c4 - c3 x - a1′[x]) + Y1 a1″[x] - b1″[x]) + Y1 (2 b1′[x] - a1^(3)[x]) + b1^(3)[x]

```
eq3 = Coefficient[SymmCond, Y1^2]
```

c4 + c3 x + a1′[x]

```
DSolve[eq3 == 0, a1, x]
```

$\{\{a1 \to \text{Function}[\{x\}, -c4\,x - \frac{c3\,x^2}{2} + C[1]]\}\}$

```
a1[x_] = -c4 x - c3 x^2/2 + c5;
```

$y''' = yy'' - (y')^2$

```
SymmCond = Collect[SymmCond, {Y0, Y1, Y2}]
```

Y2 (6 c3 - b1[x]) + 2 Y1 b1′[x] + Y0 (-c3 Y1 - b1″[x]) + b1$^{(3)}$[x]

```
b1[x_] = 6 c3;
```

```
SymmCond = Collect[SymmCond, {Y0, Y1, Y2}]
```

- c3 Y0 Y1

```
c3 = 0;
```

$y''' = yy'' - (y')^2$

```
{ξ[x, y], η[x, y]}
```

```
{c5 - c4 x, c4 y}
```

```
{ξ[x, y], η[x, y]} /. {c4 → 1, c5 → 0}
{ξ[x, y], η[x, y]} /. {c4 → 0, c5 → 1}
```

```
{-x, y}
```

```
{1, 0}
```

Concluding Remarks For ODEs

You now know:

- What a One Parameter Group of Point Transformations is.
- How to get the infinitesimal generator from the group.
- How to find the group from the generator.
- How to find the prolongation of a generator
- How to find symmetries of ODEs.
- How to use the symmetries to reduce the order of an ODE.

Partial Differential Equations

Our unknown:

$$u = u(x, t)$$

Equation:

$$F(x, t, u, u_x, u_t) = 0$$

One Parameter Group of Point Transformations:

$$\Gamma_\epsilon : \begin{cases} \bar{x} = \bar{x}(x, t, u; \epsilon) \\ \bar{t} = \bar{t}(x, t, u; \epsilon) \\ \bar{u} = \bar{u}(x, t, u; \epsilon) \end{cases}$$

For first order PDEs:

$$X = \xi_1 \frac{\partial}{\partial x} + \xi_2 \frac{\partial}{\partial t} + \eta \frac{\partial}{\partial u}$$

Heat Equation

First prolongation:

$$X^{(1)} = X + \eta_{10}\frac{\partial}{\partial u_x} + \eta_{01}\frac{\partial}{\partial u_t}$$

Second prolongation:

$$X^{(2)} = X^{(1)} + \eta_{20}\frac{\partial}{\partial u_{xx}} + \eta_{11}\frac{\partial}{\partial u_{xt}} + \eta_{02}\frac{\partial}{\partial u_{tt}}$$

Heat Equation

$U_t = U_{xx}$

```
F = UT - UXX
ux = D[u[x, t], x];
ut = D[u[x, t], t];
utx = D[ut, x];
uxx = D[ux, x];
```

```
SymmetryCondition = η01 D[F, UT] + η20 D[F, UXX]
```

$\eta 01 - \eta 20$

Heat Equation

Enter our prolongation formula:

```
η01 = D[η[x, t, u[x, t]], t] - ux D[ξ1[x, t, u[x, t]], t] - ut D[ξ2[x, t, u[x, t]], t]
η10 = D[η[x, t, u[x, t]], x] - ux D[ξ1[x, t, u[x, t]], x] - ut D[ξ2[x, t, u[x, t]], x]
η20 = D[η10, x] - uxx D[ξ1[x, t, u[x, t]], x] - utx D[ξ2[x, t, u[x, t]], x]
```

$$\begin{aligned}
&u^{(0,1)}[x,t]\,\eta^{(0,0,1)}[x,t,u[x,t]] + \eta^{(0,1,0)}[x,t,u[x,t]] - \\
&\quad u^{(1,0)}[x,t]\left(u^{(0,1)}[x,t]\,\xi 1^{(0,0,1)}[x,t,u[x,t]] + \xi 1^{(0,1,0)}[x,t,u[x,t]]\right) - \\
&\quad u^{(0,1)}[x,t]\left(u^{(0,1)}[x,t]\,\xi 2^{(0,0,1)}[x,t,u[x,t]] + \xi 2^{(0,1,0)}[x,t,u[x,t]]\right)
\end{aligned}$$

$$\begin{aligned}
&u^{(1,0)}[x,t]\,\eta^{(0,0,1)}[x,t,u[x,t]] + \eta^{(1,0,0)}[x,t,u[x,t]] - \\
&\quad u^{(1,0)}[x,t]\left(u^{(1,0)}[x,t]\,\xi 1^{(0,0,1)}[x,t,u[x,t]] + \xi 1^{(1,0,0)}[x,t,u[x,t]]\right) - \\
&\quad u^{(0,1)}[x,t]\left(u^{(1,0)}[x,t]\,\xi 2^{(0,0,1)}[x,t,u[x,t]] + \xi 2^{(1,0,0)}[x,t,u[x,t]]\right)
\end{aligned}$$

$$\begin{aligned}
&u^{(2,0)}[x,t]\,\eta^{(0,0,1)}[x,t,u[x,t]] - 2\,u^{(2,0)}[x,t]\left(u^{(1,0)}[x,t]\,\xi 1^{(0,0,1)}[x,t,u[x,t]] + \xi 1^{(1,0,0)}[x,t,u[x,t]]\right) - \\
&\quad 2\,u^{(1,1)}[x,t]\left(u^{(1,0)}[x,t]\,\xi 2^{(0,0,1)}[x,t,u[x,t]] + \xi 2^{(1,0,0)}[x,t,u[x,t]]\right) + u^{(1,0)}[x,t]\,\eta^{(1,0,1)}[x,t,u[x,t]] + \\
&\quad u^{(1,0)}[x,t]\left(u^{(1,0)}[x,t]\,\eta^{(0,0,2)}[x,t,u[x,t]] + \eta^{(1,0,1)}[x,t,u[x,t]]\right) + \eta^{(2,0,0)}[x,t,u[x,t]] - \\
&\quad u^{(1,0)}[x,t]\Big(u^{(2,0)}[x,t]\,\xi 1^{(0,0,1)}[x,t,u[x,t]] + u^{(1,0)}[x,t]\,\xi 1^{(1,0,1)}[x,t,u[x,t]] + \\
&\qquad u^{(1,0)}[x,t]\left(u^{(1,0)}[x,t]\,\xi 1^{(0,0,2)}[x,t,u[x,t]] + \xi 1^{(1,0,1)}[x,t,u[x,t]]\right) + \xi 1^{(2,0,0)}[x,t,u[x,t]]\Big) - \\
&\quad u^{(0,1)}[x,t]\Big(u^{(2,0)}[x,t]\,\xi 2^{(0,0,1)}[x,t,u[x,t]] + u^{(1,0)}[x,t]\,\xi 2^{(1,0,1)}[x,t,u[x,t]] + \\
&\qquad u^{(1,0)}[x,t]\left(u^{(1,0)}[x,t]\,\xi 2^{(0,0,2)}[x,t,u[x,t]] + \xi 2^{(1,0,1)}[x,t,u[x,t]]\right) + \xi 2^{(2,0,0)}[x,t,u[x,t]]\Big)
\end{aligned}$$

Heat Equation

Again, we try to look at this almost algebraically so we do the following substitutions for our code:

```
SymmetryCondition = SymmetryCondition /. {ux → UX, ut → UT, utx → UTX, uxx → UXX, u[x, t] → U}
```

$$
\begin{aligned}
&UT\,\eta^{(0,0,1)}[x,t,U] - UXX\,\eta^{(0,0,1)}[x,t,U] + \eta^{(0,1,0)}[x,t,U] - \\
&\quad UX\left(UT\,\xi1^{(0,0,1)}[x,t,U] + \xi1^{(0,1,0)}[x,t,U]\right) - UT\left(UT\,\xi2^{(0,0,1)}[x,t,U] + \xi2^{(0,1,0)}[x,t,U]\right) + \\
&\quad 2\,UXX\left(UX\,\xi1^{(0,0,1)}[x,t,U] + \xi1^{(1,0,0)}[x,t,U]\right) + 2\,UTX\left(UX\,\xi2^{(0,0,1)}[x,t,U] + \xi2^{(1,0,0)}[x,t,U]\right) - \\
&\quad UX\,\eta^{(1,0,1)}[x,t,U] - UX\left(UX\,\eta^{(0,0,2)}[x,t,U] + \eta^{(1,0,1)}[x,t,U]\right) - \eta^{(2,0,0)}[x,t,U] + \\
&\quad UX\left(UXX\,\xi1^{(0,0,1)}[x,t,U] + UX\,\xi1^{(1,0,1)}[x,t,U] + UX\left(UX\,\xi1^{(0,0,2)}[x,t,U] + \xi1^{(1,0,1)}[x,t,U]\right) + \xi1^{(2,0,0)}[x,t,U]\right) + \\
&\quad UT\left(UXX\,\xi2^{(0,0,1)}[x,t,U] + UX\,\xi2^{(1,0,1)}[x,t,U] + UX\left(UX\,\xi2^{(0,0,2)}[x,t,U] + \xi2^{(1,0,1)}[x,t,U]\right) + \xi2^{(2,0,0)}[x,t,U]\right)
\end{aligned}
$$

Heat Equation

Our equation allows the substitution:

```
SymmetryCondition = SymmetryCondition /. UXX → UT
```

$$\eta^{(0,1,0)}[x,t,U] - UX\left(UT\,\xi1^{(0,0,1)}[x,t,U] + \xi1^{(0,1,0)}[x,t,U]\right) -$$
$$UT\left(UT\,\xi2^{(0,0,1)}[x,t,U] + \xi2^{(0,1,0)}[x,t,U]\right) +$$
$$2\,UT\left(UX\,\xi1^{(0,0,1)}[x,t,U] + \xi1^{(1,0,0)}[x,t,U]\right) +$$
$$2\,UTX\left(UX\,\xi2^{(0,0,1)}[x,t,U] + \xi2^{(1,0,0)}[x,t,U]\right) -$$
$$UX\,\eta^{(1,0,1)}[x,t,U] - UX\left(UX\,\eta^{(0,0,2)}[x,t,U] + \eta^{(1,0,1)}[x,t,U]\right) -$$
$$\eta^{(2,0,0)}[x,t,U] + UX\left(UT\,\xi1^{(0,0,1)}[x,t,U] + UX\,\xi1^{(1,0,1)}[x,t,U] +\right.$$
$$\left.UX\left(UX\,\xi1^{(0,0,2)}[x,t,U] + \xi1^{(1,0,1)}[x,t,U]\right) + \xi1^{(2,0,0)}[x,t,U]\right) +$$
$$UT\left(UT\,\xi2^{(0,0,1)}[x,t,U] + UX\,\xi2^{(1,0,1)}[x,t,U] +\right.$$
$$\left.UX\left(UX\,\xi2^{(0,0,2)}[x,t,U] + \xi2^{(1,0,1)}[x,t,U]\right) + \xi2^{(2,0,0)}[x,t,U]\right)$$

Heat Equation

Collecting gives us this look:

```
In[16]:= SymmetryCondition = Collect[SymmetryCondition, {UX, UT, UTX}]
```

Out[16]=

$$UX^3\,\xi 1^{(0,0,2)}[x,t,U] + \eta^{(0,1,0)}[x,t,U] + 2\,UTX\,\xi 2^{(1,0,0)}[x,t,U] + UX^2\left(-\eta^{(0,0,2)}[x,t,U] + UT\,\xi 2^{(0,0,2)}[x,t,U] + 2\,\xi 1^{(1,0,1)}[x,t,U]\right) - \eta^{(2,0,0)}[x,t,U] + UX\left(2\,UTX\,\xi 2^{(0,0,1)}[x,t,U] - \xi 1^{(0,1,0)}[x,t,U] - 2\,\eta^{(1,0,1)}[x,t,U] + UT\left(2\,\xi 1^{(0,0,1)}[x,t,U] + 2\,\xi 2^{(1,0,1)}[x,t,U]\right) + \xi 1^{(2,0,0)}[x,t,U]\right) + UT\left(-\xi 2^{(0,1,0)}[x,t,U] + 2\,\xi 1^{(1,0,0)}[x,t,U] + \xi 2^{(2,0,0)}[x,t,U]\right)$$

Heat Equation

```
DeterminingEquations = DeleteCases[
  CoefficientList[SymmetryCondition, {UX, UT, UTX}] // Flatten, 0, {-1}]
```

Out[20]= $\{\eta^{(0,1,0)}[x,t,U]-\eta^{(2,0,0)}[x,t,U],\ 2\,\xi2^{(1,0,0)}[x,t,U],$
$-\xi2^{(0,1,0)}[x,t,U]+2\,\xi1^{(1,0,0)}[x,t,U]+\xi2^{(2,0,0)}[x,t,U],$
$-\xi1^{(0,1,0)}[x,t,U]-2\,\eta^{(1,0,1)}[x,t,U]+\xi1^{(2,0,0)}[x,t,U],$
$2\,\xi2^{(0,0,1)}[x,t,U],\ 2\,\xi1^{(0,0,1)}[x,t,U]+2\,\xi2^{(1,0,1)}[x,t,U],$
$-\eta^{(0,0,2)}[x,t,U]+2\,\xi1^{(1,0,1)}[x,t,U],\ \xi2^{(0,0,2)}[x,t,U],\ \xi1^{(0,0,2)}[x,t,U]\}$

- We are deleting every case of zero, -1 here makes sure that zeroes from everywhere (means deepest level in mathematica, don't worry about it)
- We get the coefficients of UX, UT, UTX,
- Flatten the list to single level, zeroes removed.

Heat Equation

To see it better:

```
DeterminingEquations // MatrixForm
```

$$\begin{pmatrix} \eta^{(0,1,0)}[x,t,U] - \eta^{(2,0,0)}[x,t,U] \\ 2\,\xi 2^{(1,0,0)}[x,t,U] \\ -\xi 2^{(0,1,0)}[x,t,U] + 2\,\xi 1^{(1,0,0)}[x,t,U] + \xi 2^{(2,0,0)}[x,t,U] \\ -\xi 1^{(0,1,0)}[x,t,U] - 2\,\eta^{(1,0,1)}[x,t,U] + \xi 1^{(2,0,0)}[x,t,U] \\ 2\,\xi 2^{(0,0,1)}[x,t,U] \\ 2\,\xi 1^{(0,0,1)}[x,t,U] + 2\,\xi 2^{(1,0,1)}[x,t,U] \\ -\eta^{(0,0,2)}[x,t,U] + 2\,\xi 1^{(1,0,1)}[x,t,U] \\ \xi 2^{(0,0,2)}[x,t,U] \\ \xi 1^{(0,0,2)}[x,t,U] \end{pmatrix}$$

```
ξ1[x_, t_, U_] = A1[x, t] U + A2[x, t];
ξ2[x_, t_, U_] = B[t];
```

Heat Equation

We have started an iterative process:

```
DeterminingEquations2 = DeleteCases[CoefficientList[SymmetryCondition, {UX, UT, UTX}] // Flatten, 0, {-1}]
```

$\{\eta^{(0,1,0)}[x, t, U] - \eta^{(2,0,0)}[x, t, U], -B'[t] + 2\,(U\,A1^{(1,0)}[x, t] + A2^{(1,0)}[x, t]),$
$-U\,A1^{(0,1)}[x, t] - A2^{(0,1)}[x, t] + U\,A1^{(2,0)}[x, t] + A2^{(2,0)}[x, t] - 2\,\eta^{(1,0,1)}[x, t, U],$
$2\,A1[x, t], 2\,A1^{(1,0)}[x, t] - \eta^{(0,0,2)}[x, t, U]\}$

Heat Equation

```
DeterminingEquations2 // MatrixForm
```

$$
\begin{pmatrix}
\eta^{(0,1,0)}[x,t,U] - \eta^{(2,0,0)}[x,t,U] \\
-B'[t] + 2\left(U\,A1^{(1,0)}[x,t] + A2^{(1,0)}[x,t]\right) \\
-U\,A1^{(0,1)}[x,t] - A2^{(0,1)}[x,t] + U\,A1^{(2,0)}[x,t] + A2^{(2,0)}[x,t] - 2\,\eta^{(1,0,1)}[x,t,U] \\
2\,A1[x,t] \\
2\,A1^{(1,0)}[x,t] - \eta^{(0,0,2)}[x,t,U]
\end{pmatrix}
$$

```
A1[x_, t_] = 0;
```

Heat Equation

```
DeterminingEquations3 = DeleteCases[CoefficientList[SymmetryCondition, {UX, UT, UTX}] // Flatten, 0, {-1}]
```

$$\left\{\eta^{(0,1,0)}[x, t, U] - \eta^{(2,0,0)}[x, t, U],\ -B'[t] + 2\,A2^{(1,0)}[x, t],\ -A2^{(0,1)}[x, t] + A2^{(2,0)}[x, t] - 2\,\eta^{(1,0,1)}[x, t, U],\ -\eta^{(0,0,2)}[x, t, U]\right\}$$

```
DeterminingEquations3 // MatrixForm
```

$$\begin{pmatrix} \eta^{(0,1,0)}[x, t, U] - \eta^{(2,0,0)}[x, t, U] \\ -B'[t] + 2\,A2^{(1,0)}[x, t] \\ -A2^{(0,1)}[x, t] + A2^{(2,0)}[x, t] - 2\,\eta^{(1,0,1)}[x, t, U] \\ -\eta^{(0,0,2)}[x, t, U] \end{pmatrix}$$

```
η[x_, t_, U_] = C1[x, t] U + C2[x, t];
```

Heat Equation

```
DSolve[-B′[t] + 2 A2^(1,0)[x, t] == 0, A2, {x, t}]
```

$$\{\{A2 \to \text{Function}[\{x, t\}, C[1][t] + \frac{1}{2} x B'[t]]\}\}$$

$$A2[x\_, t\_] = A21[t] + \frac{1}{2} x B'[t];$$

Heat Equation

```
DeterminingEquations4 = DeleteCases[CoefficientList[SymmetryCondition, {UX, UT, UTX}] // Flatten, 0, {-1}]
```

$\left\{U\,C1^{(0,1)}[x,t] + C2^{(0,1)}[x,t] - U\,C1^{(2,0)}[x,t] - C2^{(2,0)}[x,t],\ -A21'[t] - \frac{1}{2}x\,B''[t] - 2\,C1^{(1,0)}[x,t]\right\}$

```
DeterminingEquations4 // MatrixForm
```

$$\begin{pmatrix} U\,C1^{(0,1)}[x,t] + C2^{(0,1)}[x,t] - U\,C1^{(2,0)}[x,t] - C2^{(2,0)}[x,t] \\ -A21'[t] - \frac{1}{2}x\,B''[t] - 2\,C1^{(1,0)}[x,t] \end{pmatrix}$$

$\text{DSolve}\left[-A21'[t] - \frac{1}{2}x\,B''[t] - 2\,C1^{(1,0)}[x,t] == 0,\ C1,\ \{x,t\}\right]$

$\left\{\left\{C1 \to \text{Function}\left[\{x,t\},\ C[1][t] + \frac{1}{4}\left(-2\,x\,A21'[t] - \frac{1}{2}x^2\,B''[t]\right)\right]\right\}\right\}$

$C1[x\_,t\_] = C11[t] + \frac{1}{4}\left(-2\,x\,A21'[t] - \frac{1}{2}x^2\,B''[t]\right);$

Heat Equation

```
SymmetryCondition = Collect[SymmetryCondition // FullSimplify, {U}]
```

$$\frac{1}{8} U \left(8\, C11'[t] + 2\, B''[t] - x \left(4\, A21''[t] + x\, B^{(3)}[t]\right)\right) + C2^{(0,1)}[x, t] - C2^{(2,0)}[x, t]$$

```
eq1 = Collect[8 Coefficient[SymmetryCondition, U], x]
```

$$8\, C11'[t] - 4\, x\, A21''[t] + 2\, B''[t] - x^2\, B^{(3)}[t]$$

```
B[t_] = c1 + c2 t + c3 t^2;
A21[t_] = c4 + c5 t;
```

```
eq1
```

$$4\, c3 + 8\, C11'[t]$$

Heat Equation

```
DSolve[eq1 == 0, C11, t]
```

$$\left\{\left\{C11 \to \text{Function}\left[\{t\}, -\frac{c3\, t}{2} + C[1]\right]\right\}\right\}$$

$$C11[t\_] = -\frac{c3\, t}{2} + c6$$

$$c6 - \frac{c3\, t}{2}$$

Heat Equation

```
Symmetries = {ξ1[x, t, u], ξ2[x, t, u], η[x, t, u]}
SymmetryCondition = SymmetryCondition // FullSimplify
```

$$\left\{c4 + c5\,t + \frac{1}{2}\,(c2 + 2\,c3\,t)\,x,\; c1 + c2\,t + c3\,t^2,\; u\left(c6 - \frac{c3\,t}{2} + \frac{1}{4}\left(-2\,c5\,x - c3\,x^2\right)\right) + C2[x, t]\right\}$$

$$C2^{(0,1)}[x, t] - C2^{(2,0)}[x, t]$$

Heat Equation

So some examples of what symmetries we have is:

$$X_1 = \frac{\partial}{\partial t} \quad X_2 = \frac{\partial}{\partial x}$$

$$X_3 = x\frac{\partial}{\partial x} + 2t\frac{\partial}{\partial t}$$

$$X_4 = t\frac{\partial}{\partial x} + \frac{ux}{2}\frac{\partial}{\partial u}$$

$$X_5 = u\frac{\partial}{\partial u}$$

$$\boxed{X_\alpha = \alpha(x,t)\frac{\partial}{\partial u}, \quad \alpha_t = \alpha_{xx}}$$

Reduction

$$(x, t; u) \longrightarrow (r, s; w)$$

$$\left.\begin{aligned} Xr &= 0 \\ Xw &= 0 \\ Xs &= 1 \end{aligned}\right\} \text{This gives } X = \frac{\partial}{\partial s}$$

New equation won't have s.

Reduction

Use:

$$X = x\frac{\partial}{\partial x} + 2t\frac{\partial}{\partial t} + u\frac{\partial}{\partial u}$$

$$xw_x + 2tw_t + uw_u = 0$$

$$\frac{dx}{x} = \frac{dt}{2t} = \frac{du}{u} = \frac{dw}{o}$$

Reduction

$$\frac{dw}{0} = \frac{du}{u} \to w = C_1$$

$$\frac{dx}{x} = \frac{du}{u} \to \frac{u}{x} = C_2$$

$$\frac{dx}{x} = \frac{dt}{2t} \to \frac{x^2}{t} = C_3$$

$$C_1 = \phi(C_2, C_3)$$

Reduction

$$w = \phi(\frac{u}{x}, \frac{x^2}{t})$$

$$r = \psi(\frac{u}{x}, \frac{x^2}{t})$$

Pick:

$$\boxed{r = \frac{x^2}{t}}$$

$$\boxed{w = \frac{u}{x}}$$

Reduction

Xs=1

$$\frac{dx}{x} = \frac{ds}{1} \rightarrow s = lnx + C_1$$

$$C_2 = \frac{u}{x} \quad C_3 = \frac{x^2}{t}$$

We pick

$$s = \ln(x)$$

Reduction

$$\left.\begin{aligned} r &= \frac{x^2}{t} \\ w &= \frac{u}{x} \\ s &= \ln(x) \end{aligned}\right\} X = \frac{\partial}{\partial s}$$

We will insert

$$u(x,t) = x.w(\frac{x^2}{t}), \quad r = \frac{x^2}{t}$$

Reduction

```
u[x_, t_] = x w[x^2/t];

ReducedEq = D[u[x, t], t] - D[u[x, t], {x, 2}] // FullSimplify
```

$$-\frac{x\left(\left(6\,t+x^2\right)w'\left[\frac{x^2}{t}\right]+4\,x^2\,w''\left[\frac{x^2}{t}\right]\right)}{t^2}$$

Reduction

```
t = x^2 / r;
ReducedEq // Simplify
```

$$-\frac{r\,((6+r)\,w'[r]+4\,r\,w''[r])}{x}$$

Reduction

```
ReducedEquation = (6 + r) w′[r] + 4 r w′′[r]
```

$(6+r)\, w'[r] + 4\, r\, w''[r]$

ReducedEquation is a second order ODE

Reduction

```
t =.;
DSolve[ReducedEquation == 0, w, r]
```

$$\left\{\left\{w \to \text{Function}\left[\{r\}, C[2] + C[1]\left(-\frac{2\, e^{-r/4}}{\sqrt{r}} - \sqrt{\pi}\, \text{Erf}\left[\frac{\sqrt{r}}{2}\right]\right)\right]\right\}\right\}$$

$$w[r\_] = k2 + k1\left(-\frac{2\, e^{-r/4}}{\sqrt{r}} - \sqrt{\pi}\, \text{Erf}\left[\frac{\sqrt{r}}{2}\right]\right)$$

$$k2 + k1\left(-\frac{2\, e^{-r/4}}{\sqrt{r}} - \sqrt{\pi}\, \text{Erf}\left[\frac{\sqrt{r}}{2}\right]\right)$$

$$u[x\_, t\_] = w\left[x^2/t\right] \text{ // FullSimplify}$$

$$k2 - \frac{2\, e^{-\frac{x^2}{4t}}\, k1}{\sqrt{\frac{x^2}{t}}} - k1\sqrt{\pi}\, \text{Erf}\left[\frac{\sqrt{\frac{x^2}{t}}}{2}\right]$$

t=.; clears the variable.

Conclusions on PDEs

- Not too different from ODEs.
- Symmetries can help you turn PDEs into ODEs.

Further reading and access to code

- You can find the code for the examples in this presentation at:
- `https://idenizgun.github.io/`
- Recommended reading:
- Hans Stephani, Differential Equations: Their Solution Using Symmetries
- Peter Olver, Applications of Lie Groups to Differential Equations